

## 3 Movimentos com vínculos

### 3.2 Pêndulo simples




Fig. 3.2-1

Outro exemplo de movimento com vínculo é um pêndulo simples. Tal pêndulo é definido como uma partícula de massa  $m$  pendurada por um fio inelástico, de massa desprezível e de comprimento  $l$ .

A figura (3.2-1) mostra este pêndulo ideal e as forças que atuam sobre sua massa. Já que a partícula está em interação com só dois corpos (Terra e fio), existem somente duas forças: o peso  $mg$  e a tensão  $T$  no fio que, como força de vínculo, mantém a partícula sobre um círculo no plano-xy (o plano da página).

No começo, a partícula é afastada lateralmente até a posição A, onde o fio faz um ângulo  $\alpha_0$  com a vertical OB.

A massa, abandonada em A, oscilará no plano da página para a esquerda e para a direita da vertical.

Decompomos  $m\mathbf{g}$  em uma componente radial  $\mathbf{F}_r = m g \cos \alpha \cdot \mathbf{r}^o$  e uma componente tangencial  $\mathbf{F}_t = - m g \sin \alpha \cdot \mathbf{t}^o$ . Os vetores  $\mathbf{r}^o$  e  $\mathbf{t}^o$  são vetores unitários ortogonais na direção do raio e da tangente.

O sinal negativo aparece na expressão de  $\mathbf{F}_t$  porque ela atua sempre no sentido contrário ao deslocamento (= arco)  $s = OA$ .

(O ângulo  $\alpha$  em radianos é definido, como se sabe do secundário, como arco/ raio:  $\alpha := s/r$ . A razão  $\Delta \alpha / \Delta t$  é a velocidade angular  $\omega$ . Entre esta velocidade e a velocidade escalar  $v$  existe a relação  $\mathbf{v} = \mathbf{r} \cdot \omega$ . Pois:  $v = \Delta s / \Delta t = r \cdot \Delta \alpha / \Delta t = r \cdot \omega$ )

A aceleração tangencial, no caso do movimento circular, é

$$\mathbf{a}_t = r \cdot \frac{d^2 \alpha}{dt^2} = l \cdot \frac{d^2 \alpha}{dt^2} \quad (3.2-1)$$

onde  $l$  é o comprimento de nosso pêndulo. A aceleração normal, ou centrípeta, é

$$\mathbf{a}_n = \frac{v^2}{r} = \frac{v^2}{l} \quad (3.2-2)$$

A equação de movimento do pêndulo simples é, portanto,

$$m\mathbf{a}_t = m \cdot l \cdot \frac{d^2 \alpha}{dt^2} = -m \cdot g \cdot \text{sen} \alpha$$

Simplificando, obtemos

$$\frac{d^2 \alpha}{dt^2} = -\frac{g}{l} \cdot \text{sen} \alpha \quad (3.2-3)$$

Não é possível resolver essa equação analiticamente, devido à presença de  $\text{sen} \alpha$ . Entretanto, se o ângulo  $\alpha$  é pequeno ( $\alpha < 5^\circ$ ), podemos aproximar  $\text{sen} \alpha$  por  $\alpha$  (expresso em radianos). Para tais oscilações pequenas, a equação (3.2-3) reduz-se a

$$\frac{d^2 \alpha}{dt^2} + \frac{g}{l} \cdot \alpha = 0 \quad (3.2-4)$$

Antes de seguir com uma análise da equação aproximada (3.2-4), vamos resolver numericamente a equação correta (3.2-3) por meio da função `numeric::odesolve2` do MuPAD que já utilizamos com êxito no capítulo 2.5.

O seguinte Programa é só uma versão cortada do programa em 2.5, pois temos, agora, uma só equação diferencial de segundo grau. Primeiro fazemos um gráfico do ângulo  $\alpha$  com o tempo. Com `float(Y(1)[1]*180/PI);Y(1)[2]*1` obtemos o ângulo de deslocamento e a velocidade após 1 segundo.

Depois de um segundo, o pêndulo já passou para o outro lado da vertical OB. Sua velocidade neste momento é -2.84..m/s. Observe que  $Y(1)[2]$  é a velocidade angular  $\omega$  no instante  $t = 1$  s. Usando a relação  $v = r \cdot \omega = l \cdot \omega$ , obtemos a velocidade escalar  $v$  neste instante. Multiplicando o ângulo  $Y(1)[1]$  por  $180/PI$ , nos dá o ângulo em graus.

- `reset() :`

```
g:=9.8: l:=2:
```

```
a0:=45*PI/180://a = ângulo de deslocamento inicial
```

```
IVP:={alpha''(t)+g/l*sin(alpha)=0,alpha'(0)=0,
```

```
alpha(0)=a0}:
```

```
fields:=[alpha(t),alpha'(t)]:
```

```
ivp:=numeric::ode2vectorfield(IVP, fields):
```

```
Y := numeric::odesolve2(ivp):
```

```
float(Y(1)[1]*180/PI);Y(1)[2]*1;
```

```
-24.04521625 // alpha em graus no instante de 1s
```

```
-2.842499313 // velocidade (m/s)
```

```
//Animação
```

```
dt:=0.05:imax:=100:
```

```
plot(
```

```
plot::Point2d(t,Y(t)[1]*180/PI, Color = RGB::Blue,
```

```

VisibleFromTo = t..t + 0.99*dt,
PointSize = 2*unit::mm
) $ t in [i*dt $ i = 0..imax],
plot::Line2d([t-dt,Y(t - dt)[1]*180/PI],
[t,Y(t)[1]*180/PI], Color = RGB::Red,
VisibleAfter = t)
$ t in [i*dt $ i = 1..imax])

```




Fig. 3.2-2

O eixo y representa o ângulo em graus, o eixo x mostra o tempo em segundos.

Para determinar o **Período T** do movimento, devemos medir a separação temporal de dois pontos nos quais o movimento se repete. Observando o gráfico, parece que T é perto de 3 s. Para obter um valor mais exato para T, calculamos dois pontos no eixo do tempo x, separados por o intervalo T:

```
t2:=numeric::fsolve(Y(t)[1]=0,t=3..4):
```

```
t1:=numeric::fsolve(Y(t)[1]=0,t=0..1):
```

```
periodo:=t2[1]-t1[1]
```

```
0 = 2.951916278
```

O período da oscilação é, então, de  $T = 2.9519(16278)$  s.

De especial interesse é ver, junto com a curva anterior, também uma curva que mostra a velocidade em cada instante. Para obter gráficos de tamanhos comparáveis, multiplicamos os valores da velocidade por 10. Só temos de modificar a parte gráfica do programa:

- `//Animação`

```

dt:=0.05:imax:=100:

plot(
plot::Point2d(t,Y(t)[1]*180/PI, Color = RGB::Blue,
VisibleFromTo = t..t + 0.99*dt,
PointSize = 2*unit::mm
) $ t in [i*dt $ i = 0..imax],
plot::Line2d([t-dt,Y(t - dt)[1]*180/PI],
[t,Y(t)[1]*180/PI], Color = RGB::Red,
VisibleAfter = t)
$ t in [i*dt $ i = 1..imax],
plot::Point2d(t,Y(t)[2]*20, Color = RGB::Red,
VisibleFromTo = t..t + 0.99*dt,
PointSize = 2*unit::mm
) $ t in [i*dt $ i = 0..imax],
plot::Line2d([t-dt,Y(t - dt)[2]*20],
[t,Y(t)[2]*20], Color = RGB::Green,
VisibleAfter = t)
$ t in [i*dt $ i = 1..imax])

```




Fig. 3.2-3

A velocidade -a curva em verde- começa no ponto (0,0) e tem seus extremos nos pontos nos quais o gráfico de  $\alpha -t$  passa por zero.

Pode ser instrutivo traçar um gráfico com o eixo do tempo mostrando para cima:

- **//Animação**  
`dt:=0.05:imax:=100:`  
`plot(`  
`plot::Point2d(Y(t)[1]*180/PI,t, Color = RGB::Blue,`  
`VisibleFromTo = t..t + 0.99*dt,`  
`PointSize = 2*unit::mm`  
`) $ t in [i*dt $ i = 0..imax],`  
`plot::Line2d([Y(t - dt)[1]*180/PI,t-dt],`  
`[Y(t)[1]*180/PI,t], Color = RGB::Blue,`  
`VisibleAfter = t)`

```

$ t in [i*dt $ i = 1..imax],
plot::Point2d(Y(t)[2]*20,t, Color = RGB::Red,
VisibleFromTo = t..t + 0.99*dt,
PointSize = 2*unit::mm
) $ t in [i*dt $ i = 0..imax],
plot::Line2d([Y(t - dt)[2]*20,t-dt],
[Y(t)[2]*20,t], Color = RGB::Red,
VisibleAfter = t)
$ t in [i*dt $ i = 1..imax])

```




Fig. 3.2-4

Mas, seguramente, é muito mais ilustrativo de ver como o pêndulo mesmo oscila. Para lograr este show, temos de introduzir as coordenadas  $x$ ,  $y$  da partícula:

$$x = l \cdot \sin \alpha \quad \text{e} \quad (l - y)/l = \cos \alpha, \quad \text{ou seja} \quad y = l(1 - \cos \alpha)$$

Confira a figura 3.2-6 e o seguinte programa:

```
//Animação
```

```
dt:=0.05:imax:=100:
```

```
plot(
```

```
plot::Point2d(sin(Y(t)[1])*1,1*(1-cos(Y(t)[1])),  
Color = RGB::Red,
```

```
VisibleFromTo = t..t + 0.99*dt,
```

```
PointSize = 3*unit::mm
```

```
) $ t in [i*dt $ i = 0..imax],
```

```
plot::Line2d([1*sin(Y(t)[1]),1*(1-cos(Y(t)[1]))],  
[0,1], Color = RGB::Blue,
```

```
VisibleFromTo = t..t + 0.99*dt)
```

```
$ t in [i*dt $ i = 1..imax])
```




Fig. 3.2-5



### 3.2.1 O pêndulo simples simplificado

Voltemos agora a nossa atenção para a equação (3.2-4) na forma com  $s = \alpha \cdot l$

$$\frac{d^2 s}{dt^2} = -\frac{g}{l} s \quad (3.2-5)$$

Essa é uma equação diferencial cujas soluções são funções senoidais de  $\omega \cdot t$ . Substituindo  $s$  por  $A \cdot \text{sen}(\omega t + \phi)$ , podemos verificar diretamente que essa expressão para  $s$  satisfaz a equação (3.2-5).

Logo dizemos que

$$s = A \cdot \text{sen}(\omega t + \phi) \quad (3.2-6)$$

é a solução geral da equação (3.2-5) porque tem duas constantes arbitrárias, a amplitude  $A = (s_0^2 + v_0^2/\omega^2)^{1/2}$  e a fase inicial  $\phi$  com  $\text{sen } \phi = s_0/A$  e  $\text{cos } \phi = v_0/(\omega A)$ . A grandeza  $\omega \cdot t + \phi$  é denominada fase.  $\phi$  é o valor da fase para  $t = 0$ . A função seno repete-se cada vez que o ângulo aumenta de  $2\pi$ :

$$\text{sen}(\omega t + \phi) = \text{sen}(\omega t + \phi + 2\pi) = \text{sen}(\omega (t + 2\pi/\omega) + \phi)$$

Logo, o deslocamento da partícula repete-se após um intervalo de tempo de  $T = 2\pi/\omega$

$T$  é o período do movimento oscilatório simples (movimento harmônico) do pêndulo. A frequência  $f$  de um movimento harmônico simples é igual ao número de oscilações completas por unidade de tempo (por segundo), assim  $f = 1/T$ .

Como no caso do movimento circular uniforme temos  $\omega = 2\pi f$ .

A grandeza  $\omega$  é a denominada frequência angular.

No caso do pêndulo simples resulta que  $T = 2\pi (l/g)^{1/2}$ .

$T$  depende só de  $l$  e  $g$ , e é independente da massa da partícula.

Este fato é uma consequência da proporcionalidade entre massa inercial,  $m_i$ , e massa gravitacional,  $m_g$ . confira 2.1.4. Se isso não fosse assim, teríamos a equação seguinte

$$T = 2\pi \sqrt{\frac{l \cdot m_i}{g \cdot m_g}}$$

o seja, pêndulos de diferentes materiais deveriam ter diferentes períodos. Já Newton e Bessel mostraram que isso não é o caso e que  $m_i$  e  $m_g$  são proporcionais.

O período do pêndulo simples e simplificado, e com comprimento de  $l = 2\text{m}$ , será de  $T = 2\pi (2\text{m}/9.8\text{ms}^{-2})^{1/2} = 2,838453790\text{ s}$ , confira com a tabela na página 3.2-14.

Para o pêndulo sem a aproximação de  $\text{sen } \alpha \sim \alpha$  (em radianos), obtivemos 2,9519 s, o que é uma diferença de quase 4%.

Os matemáticos lutaram muito para obter uma fórmula de maior exatidão para o período do pêndulo "real". Um método famoso utiliza a seguinte série

$$T = 2\pi \sqrt{\frac{l}{g}} \left( 1 + \frac{1}{4} \text{sen}^2 \frac{\alpha}{2} + \frac{9}{64} \text{sen}^4 \frac{\alpha}{2} + \dots \right) \quad (3.2-7)$$

Para  $l = 2\text{m}$  e  $\alpha = 45^\circ$  obtemos apenas  $T = 2,94934\text{ s}$ .

Somente para pequenas amplitudes é suficiente tomar apenas os dois primeiros termos da correção. Com um programa como MuPAD é bastante fácil de obter um período com alta exatidão, como vimos mais acima. Outro método, usando uma integral elíptica, é muito popular, veja mais adiante.

### 3.2.2 Com lápis e papel

Considere, outra vez, a equação exata (3.2-3) que escrevemos agora na forma

$$\frac{d^2 s}{dt^2} = -g \cdot \frac{\text{sen } s}{l} \quad (3.2-8)$$

Façamos as seguintes modificações

$$\frac{d^2 s}{dt^2} = \frac{dv}{dt} = \frac{dv}{ds} \cdot \frac{ds}{dt} = \frac{dv}{ds} \cdot v = \frac{d\left(\frac{1}{2}v^2\right)}{ds}$$

assim obtemos uma equação que se deixa integrar.

$$\frac{d\left(\frac{1}{2}v^2\right)}{ds} = -g \cdot \sin \frac{s}{l}$$

Integrando e multiplicando por a massa  $m$ , resulta a seguinte expressão


$$mv^2/2 - mgl \cdot \cos(s/l) = mv_0^2/2 - mgl \cdot \cos(s_0/l) = \text{const.}$$

Isso significa: enquanto a partícula está oscilando pra cá e pra lá, a grandeza (3.2-9)

$$\frac{1}{2}mv^2 - mgl \cdot \cos \frac{s}{l} \quad (3.2-9)$$

permanece constante, sendo o seu valor igual a  $mv_0^2/2 - mgl \cdot \cos(s_0/l)$ . A velocidade inicial,  $v_0$ , foi, em nosso caso, zero. O arco inicial,  $s_0$ , podemos substituir por o ângulo inicial  $\alpha_0$  (radianos), pois  $s_0 = \alpha_0 \cdot l$ . No caso, o valor numérico para  $mv_0^2/2 - mgl \cdot \cos(s_0/l)$  será de  $m \cdot 13.86 \text{ m}^2/\text{s}^2$ . A massa  $m$  não foi dada.

Encontrar uma grandeza que não se muda durante um processo, é de suma importância, é como arrancar da natureza um segredo.



O que encontramos é, na realidade, outro exemplo da lei de conservação da energia.

Fig. 3.2-6

Da figura tiramos  $\cos \alpha = \cos (s/l) = (l-y)/l$   
ou  $l \cdot \cos(s/l) = l-y$

Substituindo essa relação na expressão (3.2-9), temos

$$mv^2/2 - mg(l-y) = mv_0^2 - mg(l-y_0) = \text{const. e daí}$$

$$\frac{1}{2}mv^2 + mgy = \frac{1}{2}mv_0^2 + mgy_0 = E \quad (3.2-10)$$

Essa relação é a mencionada lei de conservação da energia. O valor  $E$  da energia total depende das interações e das condições iniciais.

A equação (3.2-9) pode ser usada para calcular a velocidade de um pêndulo simples (sem a condição de ângulos pequenos), pois nos dá

$$v^2 = v_0^2 + 2gl (\cos \alpha - \cos \alpha_0) \quad (3.2-11)$$

Substituindo os valores numéricos, obtemos  $v = 2.84249931$  m/s, um valor que coincide muito bem com o resultado calculado por MuPAD de  $-2.84249931$  m/s

Com (3.2-11) não podemos dizer nada sobre o sinal da velocidade, não podíamos deduzir que o pêndulo estava retornando para sua posição original.

Escrevendo a Eq. (3.2-11) na forma seguinte

$$v = \frac{ds}{dt} = \sqrt{v_0^2 + 2gl(\cos \alpha - \cos \alpha_0)} \quad (3.2-12)$$

notamos que, efetivamente, logramos fazer uma integração da equação original que teve a segunda derivada  $d^2s/dt^2$ .

Então, a esperança é grande, de poder encontrar um caminho que nos permite integrar também a Eq. (3.2-12), para obter explicitamente  $s$  em função de  $t$ .

Mas, infelizmente, ninguém jamais encontrou um jeito apropriado para integrar (3.2-12). Demonstra-se que tal integral não pode ser expressa em termos de funções elementares. A integral que, finalmente, deveríamos resolver tem a forma de uma integral elíptica:

$$K(k) = \int_0^{\frac{\pi}{2}} \frac{dt}{\sqrt{1 - k^2 \sin^2 t}} \quad (3.2-13)$$

Para cada valor do parâmetro  $k := \sin \alpha / 2$ , ela pode ser calculada numericamente até o grau de precisão desejado e, em seguida, podemos fazer sua tabela; é justamente isso que se tem feito, e as tabelas são fáceis de conseguir, por exemplo no livro *Tables of Functions* por Eugene Jahnke e Fritz Emde, Dover Publications, 1945. Na página 85 do "Jahnke-Emde" encontramos a tabela da função  $K(k)$ , mas o parâmetro  $k$  é definido por  $k := \sin \alpha$ .

Com o seguinte programa calculamos  $K(\sin \alpha / 2)$ ,  $K1 := K(\sin \alpha)$  e o período  $T$  para ângulos entre 0 e 45 graus.

Todas as tabelas e gráficos no livro *Tables of Functions* foram feitas manualmente ou com a ajuda de uma máquina a manivela, como a "Brunsviga". Nossas tabelas calculam-se em segundos.

- `reset()`:// cálculo da função (3.2-13)

```

perodo:=proc(a0,afin)
begin
g:=9.8:
l:=2:
DIGITS:=10:
print("Graus K K1 T (s) "):
for a from a0 to afin step 5 do
k:=sin(a*PI/360):
k1:=sin(a*PI/180):
k2:=cos(a*PI/180):
K:=numeric::int((1/sqrt(1-
k^2*sin(x)^2)),x=0..PI/2):
K1:=numeric::int((1/sqrt(1-
k1^2*sin(x)^2)),x=0..PI/2):
T:= K*sqrt(l/g)*4:
print(a,K,K1,T):
end_for:
end_proc:
perodo(0,45)

```

**Resultados com  $T = 4 \cdot K \cdot (l/g)^{1/2}$ :**

"Graus	K	K1	T (s) "
0,	1.570796327,	1.570796327,	2.83845379
5,	1.571544297,	1.573792131,	2.839805384
10,	1.573792131,	1.582842804,	2.843867256
15,	1.577551661,	1.598142002,	2.850660785
20,	1.582842804,	1.620025899,	2.860221966
25,	1.589693871,	1.648995218,	2.872601951
30,	1.598142002,	1.685750355,	2.887867858
35,	1.608233762,	1.731245176,	2.906103827
40,	1.620025899,	1.786769135,	2.927412406
45,	1.633586307,	1.854074677,	2.951916278

Os seguintes cálculos são para pequenos ângulos entre 0 e 1 grau:

"Graus	K	K1	T (s) "
0,	1.570796327,	1.570796327,	2.83845379
0.1,	1.570796626,	1.570797523,	2.838454331
0.2,	1.570797523,	1.570801112,	2.838455952
0.3,	1.570799018,	1.570807093,	2.838458654
0.4,	1.570801112,	1.570815467,	2.838462437
0.5,	1.570803803,	1.570826233,	2.8384673
0.6,	1.570807093,	1.570839392,	2.838473245
0.7,	1.570810981,	1.570854944,	2.83848027
0.8,	1.570815467,	1.570872889,	2.838488376
0.9,	1.570820551,	1.570893227,	2.838497563
1.0,	1.570826233,	1.570915958,	2.838507831

Confira com  $T = 2\pi (l/g)^{1/2} = 2.83845379$  s na página 3.2-10

